
Linear Regression by Folding

Brian Beckman

12 Feb 2018

Introduction

Linear systems appear everywhere, and, where they don't appear naturally, linear approximations abound because non-linear systems are often intractable. Examples comprise machine learning, control systems, dynamics, robotics, and many more.

Linear regression is the standard technique for estimating the parameters or coefficients of a linear system. Usually, authors sweep linear regression under the rug, presumably because readers know all about effective ways to perform it. However, I have found this presumption not to be true. Time and again, I see code that employs the normal equations directly (that's inefficient), computes matrix inverses (that's risky), or abuses neural nets (a sledgehammer) to open a walnut (linear regression).

This paper shows an efficient, reliable method for very general linear regression. The method is mathematically equivalent to a Kalman fold for static systems. It is amongst the best known methods for linear regression.

Motivating Example

Find best-fit, unknowns m , b , where $z \approx m x + b$, given known data (z_1, z_2, \dots, z_k) and (x_1, x_2, \dots, x_k) . Write this system as a matrix equation and remember the symbols Z (*observations*, known), A (*partials*, known), and Ξ (*model*, state, unknown parameters to be estimated). Rows of Z and A come in matched pairs.

$$Z_{k \times 1} = \begin{pmatrix} z_1 \\ z_2 \\ \vdots \\ z_k \end{pmatrix} = \begin{pmatrix} x_1 & 1 \\ x_2 & 1 \\ \vdots & \vdots \\ x_k & 1 \end{pmatrix} \cdot \begin{pmatrix} m \\ b \end{pmatrix} + \text{noise} = A_{k \times 2} \cdot \Xi_{2 \times 1} + \text{samples of NormalDistribution}[0, \sigma_z] \quad (1)$$

This extends to any linear system with any number of parameters and with tensors in the matrix slots.

■ Ground Truth

Make some data by sampling a line specified by ground truth, then adding some noise. Run the faked data through our estimation procedures and see how close the estimates come to the ground truth. This

procedure tests the method. In the real world, we don't know ground truth, so it's paramount that we trust the method before deploying it. Build up trust by testing.

In[1]:=

```
ClearAll[groundTruth, m, b];
groundTruth = {m, b} = {0.5, -1. / 3.};
```

■ Partials

The partials A are a (column) vector of co-vectors (row vectors). Each co-vector is the gradient of the observation model, namely of $A \cdot \Xi$, with respect to Ξ , evaluated at a specific Ξ . Gradients of vector functions are co-vectors, that is, linear transformations of vectors. This fact becomes interesting when considering the dual problem, below.

In[3]:=

```
ClearAll[nData, min, max];
nData = 119; min = -1.; max = 3.;
ClearAll[partials];
partials = Array[{\#, 1.0} &, nData, {min, max}];
Short[partials, 3]
```

Out[7]//Short=

```
{{-1., 1.}, {-0.966102, 1.}, <<115>>, {2.9661, 1.}, {3., 1.}}
```

■ Fake Data

In[8]:=

```
ClearAll[fake];
fake[n_, σ_, A_, {m_, b_}] :=
  Table[
    RandomVariate[NormalDistribution[0, σ]] + A[[i]].{m, b},
    {i, n}];
```

In[10]:=

```
ClearAll[data, noiseSigma];
noiseSigma = 0.65;
data = fake[nData, noiseSigma, partials, groundTruth];
Short[data, 3]
```

Out[13]//Short=

```
{{-1.40787, -1.4709, -1.06779, <<114>>, 1.92515, 0.672254}}
```

■ Model

Try the Wolfram built-in. The estimated m and b are reasonably close to the ground truth.

In[14]:=

```
ClearAll[model];
model = LinearModelFit[{partials[[All, 1]], data}^T, ξ, ξ];
Normal[model]
```

Out[16]:=

$$-0.503288 + 0.568383 \xi$$

Un-comment the following line to see everything Wolfram has to say about this model (it's a lot of data).

In[17]:=


```
(*Association[(#->model[#])&/@model["Properties"]]*)
```

The plot shows that Wolfram does an acceptable job of estimating the parameters m and b that define the line:

In[18]:=

```
Show[ListPlot[{partials[[All, 1]], data}^T,
Plot[{m ξ + b, model[ξ]}, {ξ, min, max}]]
```

Out[18]:=



In the following, we show several methods that are all mathematically equivalent to each other, but have vastly different operational characteristics regarding memory usage and numerical risk.

■ Normal Equations

Equation 1 can be solved for a value of Ξ that minimizes the square error $J(\Xi) \stackrel{\text{def}}{=} (Z - A \cdot \Xi)^T \cdot (Z - A \cdot \Xi)$. Because the noise $\mathcal{N}(0, \sigma)$ has zero mean (or zero *bias*), The solution turns out to be exactly what one would get from naive algebra. $A^T \cdot A$ is square. When it is invertible,

$$(A^T \cdot A)^{-1} \cdot A^T \cdot Z = \Xi \quad (2)$$

That gives the same answer as Wolfram's built-in:

In[19]:=

Inverse[partials^T.partials].partials^T.data

Out[19]=

{0.568383, -0.503288}

The matrix $(A^T \cdot A)^{-1} \cdot A^T$ is the *Moore-Penrose left pseudoinverse*. Wolfram has a built-in for it:

In[20]:=

PseudoInverse[partials].data

Out[20]=

{0.568383, -0.503288}

$(A^T \cdot A)^{-1} \cdot A^T \cdot Z$ is a very nasty computation: in memory usage, in time, and in numerical risk. Eliminate these defects with a recurrence. This recurrence is mathematically identical to a Kalman filter in parameter-estimation mode, though we do not prove that here.

Recurrence

Fold the following recurrence over Z and A:

$$\begin{aligned}\psi &\leftarrow (\Lambda + a^T \cdot a)^{-1} \cdot (a^T \cdot \zeta + \Lambda \cdot \psi) \\ \Lambda &\leftarrow (\Lambda + a^T \cdot a)\end{aligned}\tag{3}$$

where ψ is the current estimate of Ξ , a and ζ are matched rows of A and Z, and Λ accumulates $A^T \cdot A$.

Derive the recurrence as follows: Treat the estimate-so-far, ψ , as just one more observation with information matrix (proportional to inverse covariance) $\Lambda = A_{\text{so-far}}^T \cdot A_{\text{so-far}}$. The scalar *performance* or *squared error* of the estimate, so far, is $J(x) = (z - A_{\text{so-far}} \cdot x)^T \cdot (z - A_{\text{so-far}} \cdot x) = (x - \psi)^T \cdot \Lambda \cdot (x - \psi)$, where x is the unknown true parameter vector and $\Lambda = A^T \cdot A$. Adding a new observation, ζ and its corresponding partial a , increases the error by $(\zeta - a \cdot x)^T \cdot (\zeta - a \cdot x)$. Minimizing the new total error with respect to x yields the recurrence.

The initial value of ψ does not matter much, but the initial value of Λ cannot be singular. For practical purposes, any Λ_0 with terms much smaller than typical terms in $A_{\text{so-far}}^T \cdot A_{\text{so-far}}$ will do. The example below starts with $\psi_0 = (0 \ 0)^T$ and $\Lambda_0 = \begin{pmatrix} 10^{-6} & 0 \\ 0 & 10^{-6} \end{pmatrix}$.

In[21]:=

```
ClearAll[update];
update[{ψ_, Λ_}, {ξ_, a_}] :=
  With[{Π = (Λ + aT.a)},
    {Inverse[Π].(aT.ξ + Λ.ψ), Π}];
MatrixForm /@
  ({ { mBar
      bBar }, Π } = Fold[update,
    { { 0
      0 }, { 1.0*^-6 0
              0      1.0*^-6 } },
    {List /@ data, List /@ partials}T])
```

Out[23]:=

$$\left\{ \begin{pmatrix} 0.568383 \\ -0.503288 \end{pmatrix}, \begin{pmatrix} 280.356 & 119. \\ 119. & 119. \end{pmatrix} \right\}$$

The estimates **mBar** and **bBar** are exactly what we got from Wolfram's built-in.

The mappings of **List** over the data and partials convert them into column vectors, as required by the recurrences in linear-algebra form.

The final value of Λ (called Π in the code, a returned value), is $A_{\text{full}}^T \cdot A_{\text{full}} + \Lambda_0$:

In[24]:=

```
Π - partialsT.partials
```

Out[24]:=

$$\left\{ \{1. \times 10^{-6}, 0.\}, \{0., 1. \times 10^{-6}\} \right\}$$

The covariance of the estimate Ξ is $\left(\frac{n-1}{n-2}\right) * \text{Variance}[Z - A \cdot \Xi] * \Lambda^{-1}$ except for a small contribution from the a-priori covariance Λ_0 . The correction factor $\left(\frac{n-1}{n-2}\right)$ is a generalization of Bessel's correction. The 2 in $(n-2)$ in the denominator is due to the fact that we're estimating two parameters from the data (see VAN DE GEER, Least Squares Estimation, Volume 2, pp. 1041–1045, in Encyclopedia of Statistics in Behavioral Science, Eds. Brian S. Everitt & David C. Howell, Wiley, 2005). The denominator of the correction, in general, is $n - p$, where n is the number of data and p is the number of parameters being estimated.

In[25]:=

```
Inverse[partialsT.partials] *  $\frac{nData - 1}{nData - 2}$  *
  Variance[data - partials.{mBar, bBar}] // MatrixForm
```

Out[25]//MatrixForm=

$$\begin{pmatrix} 0.00259377 & -0.00259377 \\ -0.00259377 & 0.00611074 \end{pmatrix}$$

Except for the reversed order, this is the same covariance matrix that Wolfram reports:

```
In[26]:= Reverse@ (Reverse /@ model["CovarianceMatrix"]) // MatrixForm
Out[26]//MatrixForm=
```

$$\begin{pmatrix} 0.00259377 & -0.00259377 \\ -0.00259377 & 0.00611074 \end{pmatrix}$$

■ Don't Invert That Matrix

See <https://www.johndcook.com/blog/2010/01/19/dont-invert-that-matrix/>

Replace **Inverse** with **LinearSolve**:

```
In[27]:= ClearAll[update];
update[{ψ_, Δ_}, {ξ_, a_}] :=
  With[{Π = (Δ + aᵀ·a)},
    {LinearSolve[Π, (aᵀ·ξ + Δ·ψ)], Π}];
MatrixForm /@ ({ { mBar
                    bBar }, Π } = Fold[update,
    { { 0
      0 }, { 1.0*^-6 0
              0      1.0*^-6 } },
    {List /@ data, List /@ partials}ᵀ])

Out[29]= { { 0.568383
             -0.503288 }, { 280.356 119.
                           119.    119. } }
```

exactly as before.

We have eliminated memory bloat by accumulating estimate updates one observation at a time, each with its paired partial. We reduce computation time and numerical risk by solving a linear system instead of inverting a matrix.

The Dual Problem

When the model is a co-vector (row-vector), e.g., a gradient, we have the dual (transposed) problem. In that case, the observations Ω and the model Γ are now co-vectors with elements ω and γ instead of ζ and ψ . The co-partials Θ (replacing A) are now a co-vector of vectors θ . The observation equation looks like $\Omega = \Gamma \cdot \Theta$ and the error-so-far like $(x - \gamma) \cdot \Lambda \cdot (x - \gamma)^\top$, where $\Lambda = \Theta_{\text{so-far}} \cdot \Theta_{\text{so-far}}^\top$. We don't change the name of Λ because it's symmetric. Adding a new observation ω introduces new error $(\omega - x \cdot \theta) \cdot (\omega - x \cdot \theta)^\top$. Minimizing the total error yields

$$\begin{aligned} \gamma &\leftarrow (\gamma \cdot \Lambda + \omega \cdot \theta^\top) \cdot (\Lambda + \theta \cdot \theta^\top)^{-1} \\ \Lambda &\leftarrow (\Lambda + \theta \cdot \theta^\top) \end{aligned} \tag{4}$$

LinearSolve operates on the transposed right-hand side of the recurrence, and we transpose the solution to get the recurrence.

In[30]:=

```
Transpose /@ List /@ partials[[1 ;; 3]]
```

Out[30]=

```
{{{-1.}, {1.}}, {{-0.966102}, {1.}}, {{-0.932203}, {1.}}}
```

In[31]:=

```
ClearAll[coUpdate];
coUpdate[{γ_, Δ_}, {ω_, θ_}] :=
  With[{Π = (Δ + θ.θᵀ)},
    {LinearSolve[Π, Δ.γᵀ + θ.ωᵀ]ᵀ, Π};
MatrixForm /@ Fold[
  coUpdate,
  {(0 0), (1.0*^-6 0; 0 1.0*^-6)},
  {List /@ List /@ data, Transpose /@ List /@ partials}ᵀ]
```

Out[31]=

```
{(0.568383 -0.503288), (280.356 119.; 119. 119.)}
```

■ Application of the Dual Problem

Imagine a scalar function $J(\theta)$ of a column K -vector $\theta_{K \times 1}$. We want to estimate its gradient co-vector ∇J , given a batch of \mathcal{I} random increments $\Delta\theta_{K \times \mathcal{I}}$, from the system $\nabla J \cdot \Delta\theta = \Delta J$. Here, ∇J takes the role of the model whose state parameters Γ we want to estimate, $\Delta\theta$ takes the role of the partials of the model w.r.t. those parameters, and ΔJ takes the role of measured data. Let $\Delta J_{\mathcal{I} \times 1}$ be a *batch* of observed increments to J and $\Delta\theta_{K \times \mathcal{I}}$ be a matrix of the \mathcal{I} corresponding column-vector random increments to the input vectors $\theta_{K \times 1}$. The right pseudoinverse $\text{RPI} \stackrel{\text{def}}{=} (\Delta\theta_{K \times \mathcal{I}} \cdot \Delta\theta_{K \times \mathcal{I}}^\top)^{-1} \cdot \Delta\theta_{K \times \mathcal{I}}^\top$ solves $\nabla J_{\mathcal{I} \times 1} \cdot \Delta\theta_{K \times \mathcal{I}} = \Delta J_{\mathcal{I} \times 1}$ to yield $\nabla J_{\mathcal{I} \times 1} \approx \Delta J_{\mathcal{I} \times 1} \cdot \text{RPI}$. Use the co-update recurrence for this problem.