Linear Regression by Folding

Brian Beckman
12 Feb 2018

Motivating Example

Find best-fit, unknowns m, b, where $z \approx m\,x + b$, given known data $(z_1, z_2, \ldots, z_n)$ and $(x_1, x_2, \ldots, x_n)$. Write this *system* as a matrix equation and remember the symbols Z (*observations*, known), A (*partials*, known), and Ξ (*model, state, unknown parameters to be estimated*). Rows of Z and A come in matched pairs.

$$Z_{k\times 1} = \begin{pmatrix} z_1 \\ z_2 \\ \vdots \\ z_k \end{pmatrix} = \begin{pmatrix} x_1 & 1 \\ x_2 & 1 \\ \vdots & \vdots \\ x_k & 1 \end{pmatrix} \cdot \begin{pmatrix} m \\ b \end{pmatrix} + \text{noise} = A_{k\times 2} \cdot \Xi_{2\times 1} + \text{samples of NormalDistribution}[0, \sigma_Z] \tag{1}$$

This extends to any linear system with any number of parameters and with tensors in the matrix slots.

Ground Truth

Make some data by sampling a line specified by the following ground truth, then adding some noise. Run the faked data through our estimation procedures and see how close the estimates come to the ground truth.

In[1]:=
```
ClearAll[groundTruth, m, b];
groundTruth = {m, b} = {0.5, -1. / 3.};
```

Partials

The partials A are a (column) vector of co-vectors (row vectors). Each co-vector is the gradient of the observation model, namely of $A \cdot \Xi$, with respect to Ξ, evaluated at a specific Ξ. Gradients of vector functions are co-vectors, that is, linear transformations of vectors. This fact becomes interesting when considering the dual problem, below.

In[3]:=
```
ClearAll[nData, min, max];
nData = 119; min = -1.; max = 3.;
ClearAll[partials];
partials = Array[{#, 1.0} &, nData, {min, max}];
Short[partials, 3]
```

Out[7]//Short=
```
{{-1., 1.}, {-0.966102, 1.}, {-0.932203, 1.},
 {-0.898305, 1.}, <<112>>, {2.9322, 1.}, {2.9661, 1.}, {3., 1.}}
```

Fake Data

In[8]:=
```
ClearAll[fake];
fake[n_, σ_, A_, {m_, b_}] :=
  Table[
   RandomVariate[NormalDistribution[0, σ]] + A[[i]].{m, b},
   {i, n}];
```

In[10]:=
```
ClearAll[data, noiseSigma];
noiseSigma = 0.65;
data = fake[nData, noiseSigma, partials, groundTruth];
Short[data, 3]
```

Out[13]//Short=
```
{-1.06119, 0.0515971, -1.31516, 0.162079,
 <<111>>, 1.5034, 1.17977, 1.76527, 1.97062}
```

Model

Try the Wolfram built-in. The estimated *m* and *b* are reasonably close to the ground truth.

In[14]:=
```
ClearAll[model];
model = LinearModelFit[{partials[[All, 1]], data}^T, ξ, ξ];
Normal[model]
```

Out[16]=
$$-0.28379 + 0.496262\,\xi$$

Un-comment the following line to see everything Wolfram has to say about this model (it's a lot of data).

In[17]:=
```
(*Association[(#→model[#])&/@model["Properties"]]*)
```

```
Show[ListPlot[{partials[[All, 1]], data}ᵀ],
 Plot[{m ξ + b, model[ξ]}, {ξ, min, max}]]
```

Out[18]=



Normal Equations

Equation 1 can be solved for a value of Ξ that minimizes the square error $J(\Xi) \stackrel{\text{def}}{=} (Z - A\cdot\Xi)^{\intercal}\cdot(Z - A\cdot\Xi)$. Because the noise $\mathcal{N}(0, \sigma)$ has zero mean (or zero *bias*), The solution turns out to be exactly what one would get from naive algebra. $A^{\intercal}\cdot A$ is square. When it is invertible,

$$(A^{\intercal}\cdot A)^{-1}\cdot A^{\intercal}\cdot Z = \Xi \tag{2}$$

That gives the same answer as Wolfram's built-in:

In[19]:=
```
Inverse[partialsᵀ.partials].partialsᵀ.data
```

Out[19]=
```
{0.496262, -0.28379}
```

The matrix $(A^{\intercal}\cdot A)^{-1}\cdot A^{\intercal}$ is the *Moore-Penrose left pseudoinverse*. Wolfram has a built-in for it:

In[20]:=
```
PseudoInverse[partials].data
```

Out[20]=
```
{0.496262, -0.28379}
```

$(A^{\intercal}\cdot A)^{-1}\cdot A^{\intercal}\cdot Z$ is a very nasty computation, in memory usage, in time, and in numerical risk. Eliminate these defects with a recurrence. This recurrence is mathematically identical to a Kalman filter in parameter-estimation mode, though we do not prove that here.

Recurrence

Fold the following recurrence over Z and *A*:

$$\psi \leftarrow (\Lambda + a^{\intercal}\cdot a)^{-1}\cdot(a^{\intercal}\cdot\zeta + \Lambda\cdot\psi)$$
$$\Lambda \leftarrow (\Lambda + a^{\intercal}\cdot a) \tag{3}$$

where ψ is the current estimate of Ξ, a and ζ are matched rows of A and Z, and Λ accumulates $A^{\intercal}\cdot A$.

Derive the recurrence as follows: Treat the estimate-so-far, ψ, as just one more observation with information matrix (proportional to inverse covariance) $\Lambda = A_{so\text{-}far}{}^{\intercal}\cdot A_{so\text{-}far}$. The scalar *performance* or *squared error* of the estimate, so far, is $J(x) = (z - A_{so\text{-}far}\cdot x)^{\intercal}\cdot(z - A_{so\text{-}far}\cdot x) = (x-\psi)^{\intercal}\cdot\Lambda\cdot(x-\psi)$, where x is the unknown true parameter vector and $\Lambda = A^{\intercal}\cdot A$. Adding a new observation, ζ and its corresponding partial a, increases the error by $(\zeta - a\cdot x)^{\intercal}\cdot(\zeta - a\cdot x)$. Minimizing the new total error with respect to x yields the recurrence.

The initial value of ψ does not matter much, but the initial value of Λ cannot be singular. For practical purposes, any Λ_0 with terms much smaller than typical terms in $A_{so\text{-}far}{}^{\intercal}\cdot A_{so\text{-}far}$ will do. The example below starts with $\psi_0 = (\,0\quad 0\,)^{\intercal}$ and $\Lambda_0 = \left(\begin{pmatrix}10^{-6} & 0\end{pmatrix}\begin{pmatrix}0 & 10^{-6}\end{pmatrix}\right)$.

In[76]:=
```
ClearAll[update];
update[{ψ_, Λ_}, {ζ_, a_}] :=
  With[{Π = (Λ + a^T.a)},
   {Inverse[Π].(a^T.ζ + Λ.ψ), Π}];
MatrixForm /@
 ({{mBar, bBar}, Π} = Fold[update,
    {{0, 0}, {{1.0*^-6, 0}, {0, 1.0*^-6}}},
    {List /@ data, List /@ partials}^T])
```

Out[78]=
$$\left\{\begin{pmatrix}0.496262\\ -0.28379\end{pmatrix}, \begin{pmatrix}280.356 & 119.\\ 119. & 119.\end{pmatrix}\right\}$$

The estimates **mBar** and **bBar** are exactly what we got from Wolfram's built-in.

The mappings of **List** over the data and partials convert them into column vectors, as required by the recurrences in linear-algebra form.

The final value of Λ, called Π in the code, is $A_{full}{}^{\intercal}\cdot A_{full} + \Lambda_0$

In[24]:=
```
Π - partials^T.partials
```

Out[24]=
```
{{1. × 10^-6, 0.}, {0., 1. × 10^-6}}
```

The covariance of the estimate Ξ is $(\frac{n-1}{n-2})$*Variance[Z – A·Ξ]*Λ^{-1} except for a small contribution from the a-priori covariance Λ_0. The correction factor $(\frac{n-1}{n-2})$ is a generalization of Bessel's correction. The 2 in (*n* – 2) in the denominator is due to the fact that we're estimating two parameters from the data (see VAN DE GEER, Least Squares Estimation, Volume 2, pp. 1041–1045, in Encyclopedia of Statistics in Behavioral Science, Eds. Brian S. Everitt & David C. Howell, Wiley, 2005). The denominator of the correction, in general, is *n* – *p*, where *n* is the number of data and *p* is the number of parameters being estimated.

In[62]:=
```
Inverse[partials^T.partials] * (nData - 1)/(nData - 2) *
  Variance[data - partials.{mBar, bBar}] // MatrixForm
```

Out[62]//MatrixForm=
$$\begin{pmatrix} 0.00236492 & -0.00236492 \\ -0.00236492 & 0.00557159 \end{pmatrix}$$

Except for the reversed order, this is the same covariance matrix that Wolfram reports:

In[26]:=
```
Reverse@(Reverse /@ model["CovarianceMatrix"]) // MatrixForm
```

Out[26]//MatrixForm=
$$\begin{pmatrix} 0.00236492 & -0.00236492 \\ -0.00236492 & 0.00557159 \end{pmatrix}$$

Don't Invert That Matrix

See https://www.johndcook.com/blog/2010/01/19/dont-invert-that-matrix/

Replace **Inverse** with **LinearSolve**:

In[79]:=
```
ClearAll[update];
update[{ψ_, Λ_}, {ζ_, a_}] :=
  With[{Π = (Λ + aᵀ.a)},
   {LinearSolve[Π, (aᵀ.ζ + Λ.ψ)], Π}];
MatrixForm /@ ({{mBar, bBar}, Π} = Fold[update,
    {{0, 0}, {{1.0*^-6, 0}, {0, 1.0*^-6}}},
    {List /@ data, List /@ partials}ᵀ])
```

Out[81]=
$$\left\{\begin{pmatrix} 0.496262 \\ -0.28379 \end{pmatrix}, \begin{pmatrix} 280.356 & 119. \\ 119. & 119. \end{pmatrix}\right\}$$

exactly as before.

We have eliminated memory bloat by accumulating estimate updates one observation at a time, with its paired partial. We reduce computation time and numerical risk by solving a linear system instead of inverting a matrix.

The Dual Problem

When the model is a co-vector (row-vector), e.g., a gradient, we have the dual (transposed) problem. In that case, the observations Ω and the model Γ are now co-vectors with elements ω and γ instead of ζ and ψ. The co-partials Θ (replacing A) are now a co-vector of vectors θ. The observation equation looks like $\Omega = \Gamma \cdot \Theta$ and the error-so-far like $(x - \gamma)\cdot\Lambda\cdot(x - \gamma)^{\intercal}$, where $\Lambda = \Theta_{so\text{-}far}\cdot\Theta^{\intercal}_{so\text{-}far}$. We don't need to change the name of Λ because it's symmetric. Adding a new observation ω introduces new error $(\omega - x\cdot\theta)\cdot(\omega - x\cdot\theta)^{\intercal}$. Minimizing the total error yields

$$\begin{aligned} \gamma &\leftarrow (\gamma\cdot\Lambda + \omega\cdot\theta^{\intercal}).(\Lambda + \theta\cdot\theta^{\intercal})^{-1} \\ \Lambda &\leftarrow (\Lambda + \theta\cdot\theta^{\intercal}) \end{aligned} \tag{4}$$

LinearSolve operates on the transposed right-hand side of the recurrence, and we transpose the solution to get the recurrence.

In[109]:=
```
Transpose /@ List /@ partials[[1 ;; 3]]
```

Out[109]=
```
{{{-1.}, {1.}}, {{-0.966102}, {1.}}, {{-0.932203}, {1.}}}
```

In[142]:=
```
ClearAll[coUpdate];
coUpdate[{γ_, Λ_}, {ω_, Θ_}] :=
  With[{Π = (Λ + Θ.Θ^T)},
   {LinearSolve[Π, Λ.γ^T + Θ.ω^T]^T, Π}];
MatrixForm /@ Fold[
  coUpdate,
  {(0  0), (1.0*^-6     0
               0     1.0*^-6)},
  {List /@ List /@ data, Transpose /@ List /@ partials}^T]
```

Out[144]=
$$\left\{\begin{pmatrix} 0.496262 & -0.28379 \end{pmatrix}, \begin{pmatrix} 280.356 & 119. \\ 119. & 119. \end{pmatrix}\right\}$$

Application of the Dual Problem

Imagine a scalar function $J(\theta)$ of a column K-vector $\theta_{K\times 1}$. We want to estimate its gradient co-vector ∇J, given a batch of $\mathcal{I}$ random increments $\Delta\Theta_{K\times\mathcal{I}}$, from the system $\nabla J\cdot\Delta\Theta = \Delta J$. Here, ∇J takes the role of the model whose state parameters ψ we want to estimate, $\Delta\Theta$ takes the role of the partials of the model w.r.t. those parameters, and ΔJ takes the role of measured data. Let $\Delta J_{\mathcal{I}\times 1}$ be a *batch* of observed increments to J and $\Delta\Theta_{K\times\mathcal{I}}$ be a matrix of the $\mathcal{I}$ corresponding column-vector random increments to the input vectors $\theta_{K\times 1}$. The right pseudoinverse $\text{RPI} \overset{\text{def}}{=} (\Delta\Theta_{K\times\mathcal{I}}\cdot\Delta\Theta_{K\times\mathcal{I}}^{\intercal})^{-1}\cdot\Delta\Theta_{K\times\mathcal{I}}^{\intercal}$ solves $\nabla J_{\mathcal{I}\times 1}\cdot\Delta\Theta_{K\times\mathcal{I}} = \Delta J_{\mathcal{I}\times 1}$ to yield $\nabla J_{\mathcal{I}\times 1} \approx \Delta J_{\mathcal{I}\times 1}\cdot\text{RPI}$.